

ANALISI DELLA VARIANZA

In questo documento presentiamo alcune opzioni analitiche della procedura di analisi fattoriale di SPSS che non sono state incluse nel testo pubblicato. Si tratta di opzioni che, pur non essendo utilizzate correntemente per effettuare l'analisi della varianza, consentono di completare lo spettro delle possibilità di analisi offerte da SPSS. In particolare, il documento completa la descrizione della procedura di analisi fattoriale con i seguenti paragrafi:

1. Analisi degli effetti semplici dell'esempio 2 del testo
2. L'analisi del trend (analisi degli andamenti)
3. L'ANOVA con il linguaggio SINTASSI
4. Opzioni aggiuntive di GLM

Modello

Salva

1. Analisi degli effetti semplici dell'esempio 2

Nel testo abbiamo presentato l'analisi degli effetti semplici relativa all'esempio 2 considerando il fattore Compito nei diversi livelli del fattore Rinforzo. Qui presentiamo gli effetti semplici relativi al fattore Rinforzo nei diversi livelli del fattore Compito. Questi effetti sono ottenibili tramite le seguenti linee di comandi SINTASSI.

```
* Analisi della varianza fattoriale "Between-subjects":  
Effetti semplici per il fattore Rinforzo.
```

```
UNIANOVA  risposte  BY  rinforzo  compito  
/METHOD = SSTYPE(3)  
/INTERCEPT = INCLUDE  
/EMMEANS = TABLES(rinforzo*compito)  
  COMPARE (RINFORZO) ADJ(SIDAK)  
/PRINT = DESCRIPTIVE ETASQ OPOWER  
/CRITERIA = ALPHA(.05)  
/DESIGN = rinforzo compito rinforzo*compito .
```

I risultati di questa procedura relativi agli effetti semplici, ottenuti tramite la linea di comando **/EMMEANS = TABLES(rinforzo*compito) COMPARE (RINFORZO) ADJ(SIDAK)**, vengono riportati nella tabella 1. I confronti a coppie tra le medie delle condizioni di Rinforzo per ogni livello di Compito (“Confronti a coppie”) sono ora necessari per interpretare i risultati degli effetti semplici poiché i livelli di Rinforzo sono più di due. I confronti vengono effettuati apportando una correzione del livello di significatività secondo l’approccio di Sidak, per limitare l’inflazione della probabilità di commettere errore del primo tipo (in particolare, la correzione di Sidak operata sul livello alfa è la seguente: $\alpha \text{ di Sidak} = 1 - (1 - \alpha)^{1/C}$ dove C è il numero di tutti i possibile confronti di coppie diverse di medie). Infine abbiamo le due tavole dei risultati generali dei disegni ANOVA monofattoriali che esaminano proprio gli effetti semplici (“Test univariati”), considerando le differenze tra i 3 livelli della variabile Rinforzo all’interno dei due livelli della variabile compito. Sia per i compiti Semplici sia per quelli complessi emerge una differenza significativa tra i Rinforzi Lode, Critica e Silenzio, e questa differenza non solo risulta significativa ma anche associata ad una dimensione dell’effetto piuttosto grande (il coefficiente eta è infatti uguale a .266 per i compiti Semplici e a .448 per quelli Complessi). Questo apparentemente sembra discordante con la presenza di un’interazione significativa, ma se osserviamo la tabella dei confronti a coppie ci accorgiamo che per i compiti Semplici l’unica differenza significativa è quella tra Lode (7.600) e Silenzio (4.400), mentre quella con la Critica (7.200) e quella tra Critica e Silenzio non risultano significative. Se invece consideriamo i compiti Complessi la Lode (7.000) non solo risulta maggiore della Critica (2.000) ma anche del Silenzio (3.200), mentre questi ultimi due continuano a non presentare differenze significative.

La devianza Between che viene scomposta nei due test univariati e che viene presentata nella tabella 1 ($30.400+68.133 = 98.533$) è data dalla somma della devianza Between del fattore “Rinforzo” (67.267) più la devianza Between dell’interazione “Rinforzo X Compito” (31.267), con due gradi di libertà perché si confrontano tre gruppi (vedi la tabella 3.12 relativa al test omnibus,

alla pag. 138 del testo). La devianza Within è sempre quella del disegno fattoriale completo (84.00), così come i gradi di libertà ad essa associati.

Tabella 1. Risultati del test degli effetti semplici

Confronti a coppie

Variabile dipendente: RISPOSTE

COMPITO	(I) RINFORZO	(J) RINFORZO	Differenza fra medie (I-J)	Errore std.	Sig. ^a	Intervallo di confidenza per la differenza al 95% ^a	
						Limite inferiore	Limite superiore
1,00 SEMPLICI	1,00 LODE	2,00 CRITICA	,400	1,183	,982	-2,636	3,436
		3,00 SILENZIO	3,200*	1,183	,037	,164	6,236
	2,00 CRITICA	1,00 LODE	-,400	1,183	,982	-3,436	2,636
		3,00 SILENZIO	2,800	1,183	,077	-,236	5,836
	3,00 SILENZIO	1,00 LODE	-3,200*	1,183	,037	-6,236	-,164
		2,00 CRITICA	-2,800	1,183	,077	-5,836	,236
2,00 COMPLESSI	1,00 LODE	2,00 CRITICA	5,000*	1,183	,001	1,964	8,036
		3,00 SILENZIO	3,800*	1,183	,011	,764	6,836
	2,00 CRITICA	1,00 LODE	-5,000*	1,183	,001	-8,036	-1,964
		3,00 SILENZIO	-1,200	1,183	,686	-4,236	1,836
	3,00 SILENZIO	1,00 LODE	-3,800*	1,183	,011	-6,836	-,764
		2,00 CRITICA	1,200	1,183	,686	-1,836	4,236

Basato sulle medie marginali stimate

*. La differenza fra medie è significativa al livello ,05

a. Correzione per confronti multipli: Sidak.

Test univariati

Variabile dipendente: RISPOSTE

COMPITO		Somma dei quadrati	df	Media dei quadrati	F	Sig.	Eta quadrato parziale	Non centralità	Potenza osservata ^a
1,00 SEMPLICI	Contrasto	30,400	2	15,200	4,343	,025	,266	8,686	,697
	Errore	84,000	24	3,500					
2,00 COMPLESSI	Contrasto	68,133	2	34,067	9,733	,001	,448	19,467	,968
	Errore	84,000	24	3,500					

F verifica l'effetto di RINFORZO. Questo test è basato sui confronti a coppie indipendenti e lineari tra le medie marginali stimate.

a. Calcolato usando alfa = ,05

Saremmo giunti a risultati analoghi se avessimo effettuato confronti pianificati invece dei confronti post-hoc, ad esempio confrontando “Lode” vs. “Critica e Silenzio” aggregati (confronto 1), e “Critica” vs. “Silenzio” (Confronto 2) nei compiti “Semplici” e nei compiti “Complessi”. I

confronti pianificati all'interno degli effetti semplici relativi al fattore Rinforzo nei diversi livelli del fattore Compito possono essere richiesti tramite la seguente sintassi:

```
* Analisi della varianza fattoriale "Between-subjects":  
Effetti semplici con confronti pianificati.  
  
UNIANOVA  risposte  BY  compito  rinforzo  
/METHOD = SSTYPE(3)  
/INTERCEPT = INCLUDE  
/lmatrix 'rinforzo at low compito level'  rinforzo 2 -1 -1  
compito*rinforzo 2 -1 -1 0 0 0  
/lmatrix 'rinforzo at low compito level'  rinforzo 0 1 -1  
compito*rinforzo 0 1 -1 0 0 0  
/lmatrix 'rinforzo at high compito level'  rinforzo 2 -1 -1  
compito*rinforzo 0 0 0 2 -1 -1  
/lmatrix 'rinforzo at high compito level'  rinforzo 0 1 -1  
compito*rinforzo 0 0 0 0 1 -1  
/PRINT = DESCRIPTIVE ETASQ OPOWER  
/CRITERIA = ALPHA(.05)  
/DESIGN = rinforzo compito rinforzo*compito .
```

I risultati di questi quattro confronti vengono presentati nella tabella 2. Nel primo confronto, effettuato nei compiti "Semplici" contrastando "Lode" e "Critica e Silenzio" aggregati, non emergono differenze significative mentre emergono nel secondo confronto che contrasta il livello "Critica" dal livello "Silenzio" sempre nei compiti Semplici. Nel terzo confronto, effettuato nei compiti "Complessi" contrastando "Lode" e "Critica e Silenzio" aggregati emerge una differenza significativa, mentre non emerge quando, sempre nei compiti Complessi, con il quarto confronto si contrastano "Critica" e "Silenzio". Dunque, a seconda dei livelli di Compito che vengono considerati, uno stesso confronto pianificata ha esiti diversi, e questo è il senso dell'interazione significativa che emerge.

Tabella 2. Confronti pianificati negli effetti semplici

```

/matrix 'rinforzo at low compito level'
rinforzo 2 -1 -1 compito*rinforzo 2 -1 -1 0 0 0
    
```

Risultati del test

Variabile dipendente: RISPOSTE

Sorgente	Somma dei quadrati	df	Media dei quadrati	F	Sig.	Eta quadrato parziale	Non centralità	Potenza osservata ^a
Contrasto	10,800	1	10,800	3,086	,092	,114	3,086	,392
Errore	84,000	24	3,500					

a. Calcolato usando alfa = ,05

```

/matrix 'rinforzo at low compito level'
rinforzo 0 1 -1 compito*rinforzo 0 1 -1 0 0 0
    
```

Risultati del test

Variabile dipendente: RISPOSTE

Sorgente	Somma dei quadrati	df	Media dei quadrati	F	Sig.	Eta quadrato parziale	Non centralità	Potenza osservata ^a
Contrasto	19,600	1	19,600	5,600	,026	,189	5,600	,622
Errore	84,000	24	3,500					

a. Calcolato usando alfa = ,05

```

/matrix 'rinforzo at high compito level'
rinforzo 2 -1 -1 compito*rinforzo 0 0 0 2 -1 -1
    
```

Risultati del test

Variabile dipendente: RISPOSTE

Sorgente	Somma dei quadrati	df	Media dei quadrati	F	Sig.	Eta quadrato parziale	Non centralità	Potenza osservata ^a
Contrasto	64,533	1	64,533	18,438	,000	,434	18,438	,984
Errore	84,000	24	3,500					

a. Calcolato usando alfa = ,05

```

/matrix 'rinforzo at high compito level'
rinforzo 0 1 -1 compito*rinforzo 0 0 0 0 1 -1
    
```

Risultati del test

Variabile dipendente: RISPOSTE

Sorgente	Somma dei quadrati	df	Media dei quadrati	F	Sig.	Eta quadrato parziale	Non centralità	Potenza osservata ^a
Contrasto	3,600	1	3,600	1,029	,321	,041	1,029	,164
Errore	84,000	24	3,500					

a. Calcolato usando alfa = ,05

2. L'analisi del trend (analisi degli andamenti)

La trend analysis (o analisi degli andamenti) viene utilizzata per esaminare ipotesi di ricerca relative alla forma della relazione che lega la variabile dipendente alla variabile indipendente. I livelli della variabile indipendente sono rappresentati da “quantità”, ovvero intervalli di tempo, dosi di un farmaco, ecc., e non da “categorie” (come Maschio e Femmina, ad esempio). La trend analysis può essere effettuata sia nei disegni “tra i soggetti” che nei disegni “entro i soggetti”, laddove si voglia esaminare ad esempio se un certo processo o una certa caratteristica si trasforma con il passare del tempo.

Le procedure per la trend analysis sono simili a quelle utilizzate per effettuare confronti pianificati: alle medie che rappresentano i diversi livelli della variabile indipendente vengono attribuiti appositi coefficienti che definiscono una particolare forma della relazione e che sono rintracciabili in molti testi sull'analisi della varianza. Per effettuare l'analisi degli andamenti è necessario che ci sia lo stesso numero di soggetti in ciascuna delle condizioni sperimentali, e che “le modalità del fattore rappresentino passi equidistanti lungo un continuum sottostante” (Ercolani, Areni e Mannetti, 1990, p. 149): ad esempio, in un disegno longitudinale l'intervallo di tempo che separa le diverse misure deve essere lo stesso. L'analisi del trend consente di scomporre la devianza dovuta al fattore in questione nella componente dovuta all'andamento (trend) lineare (l'incremento o il decremento nella variabile dipendente è costante attraverso tutti i livelli della variabile indipendente), e nelle componenti dovute ai diversi andamenti non lineari di ordine superiore (il cambiamento nella variabile dipendente non è completamente costante attraverso i diversi livelli della variabile indipendente). Poiché vengono utilizzati coefficienti che rendono i confronti ortogonali, le diverse componenti del trend non sono correlate tra di loro. Il numero di componenti che consentono di scomporre l'andamento è uguale a $k-1$, dove k è il numero di punti disponibili della variabile entro i soggetti o il numero di gruppi della variabile tra i soggetti.

Presentiamo un esempio di applicazione dell'analisi degli andamenti, relativo ad un disegno misto.

Il punteggio in una scala di aggressività è stato rilevato, tramite autovalutazione, su un campione di

208 soggetti (104 maschi e 104 femmine) in quattro momenti temporali differenti: dalla quinta elementare alla terza media. Si vuole esaminare la differenza nel punteggio di aggressività dovuta alle differenze di età (classe scolastica) e al sesso. Per effettuare l'analisi tramite SPSS bisogna seguire quanto visto nell'esempio del modello *misto* (vedi pagine 160-164 del testo): il fattore entro i soggetti sarà rappresentato dall'aggressività e avrà 4 livelli, il fattore tra i soggetti è il genere con 2 livelli. La figura 1 presenta la finestra di dialogo principale che consente di definire il fattore entro i soggetti "aggress" e di specificare il fattore tra i soggetti "sex". Il programma presenta come opzione di default per i contrasti dei fattori entro i soggetti l'opzione Polinomiale (vedi figura 2): è questa l'opzione che ci consentirà di esaminare le ipotesi relative alle componenti del trend.

Figura 1. Finestra di dialogo per definire un disegno misto




Figura 2. Finestra di dialogo dei contrasti per i fattori entro i soggetti



La tabella 3 presenta il risultato del modello ANOVA relativamente al fattore entro i soggetti e all'interazione. Solo l'effetto principale dovuto all'età (aggress) risulta significativo, anche se spiega una proporzione di varianza veramente bassa (circa il 2%). L'effetto principale del genere viene presentato nella tabella 4: si tratta infatti di un fattore tra i soggetti, che risulta significativo e che spiega inoltre una proporzione di varianza veramente molto elevata (oltre il 27%). La tabella 5 riporta le medie delle variabili che consentono di interpretare gli effetti principali: i maschi presentano un punteggio significativamente maggiore delle femmine (la media nella tabella è aggregata attraverso i 4 livelli temporali considerati); il punteggio nell'aggressività rimane stabile per i primi 3 anni e poi è soggetto ad un deciso decremento, e questo avviene sia per i maschi sia per le femmine, non essendo l'interazione significativa.

La tabella 6 riporta l'analisi del trend vera e propria che evidenzia un effetto significativo a carico della componente lineare e della componente quadratica solo nel caso dell'effetto principale. La figura 3 fornisce una rappresentazione grafica delle medie attraverso i 4 anni considerati: la componente lineare è verosimilmente responsabile della stabilità nel livello di aggressività che si manifesta attraverso i primi 3 tempi, la componente quadratica invece è responsabile del brusco decadimento che si ha passando dal tempo 3 al tempo 4. I coefficienti utilizzati dal programma per

condurre questa analisi sono i seguenti: componente lineare = -.671, -.224, .224, .671; componente quadratica = .50, -.50, -.50, .50; componente cubica = -.224, .671, -.671, .224.

Tabella 3. Risultati del modello misto relativamente al fattore entro i soggetti e all'interazione

Test degli effetti entro soggetti

Misura: MEASURE_1

Sorgente		Somma dei quadrati Tipo III	df	Media dei quadrati	F	Sig.	Eta quadrato parziale	Non centralità	Potenza osservata ^a
AGGRESS	Assumendo la sfericità	,940	3	,313	3,840	,010	,018	11,520	,821
	Greenhouse-Geisser	,940	2,879	,327	3,840	,011	,018	11,055	,809
	Huynh-Feldt	,940	2,938	,320	3,840	,010	,018	11,283	,815
	Limite inferiore	,940	1,000	,940	3,840	,051	,018	3,840	,496
AGGRESS * SEX	Assumendo la sfericità	,398	3	,133	1,625	,182	,008	4,876	,428
	Greenhouse-Geisser	,398	2,879	,138	1,625	,184	,008	4,679	,418
	Huynh-Feldt	,398	2,938	,135	1,625	,183	,008	4,775	,423
	Limite inferiore	,398	1,000	,398	1,625	,204	,008	1,625	,245
Errore(AGGRESS)	Assumendo la sfericità	50,430	618	,082					
	Greenhouse-Geisser	50,430	593,032	,085					
	Huynh-Feldt	50,430	605,248	,083					
	Limite inferiore	50,430	206,000	,245					

a. Calcolato usando alfa = ,05

Tabella 4. Risultati del modello misto relativamente al fattore tra i soggetti

Test degli effetti fra soggetti

Misura: MEASURE_1

Variabile trasformata: Media

Sorgente	Somma dei quadrati Tipo III	df	Media dei quadrati	F	Sig.	Eta quadrato parziale	Non centralità	Potenza osservata ^a
Intercept	2457,965	1	2457,965	7492,730	,000	,973	7492,730	1,000
SEX	25,519	1	25,519	77,790	,000	,274	77,790	1,000
Errore	67,578	206	,328					

a. Calcolato usando alfa = ,05

Tabella 5. Medie della variabile dipendente per i diversi livelli delle variabili indipendenti

1. SEX

Misura: MEASURE_1

SEX	Media	Errore std.	Intervallo di confidenza 95%	
			Limite inferiore	Limite superiore
1	1,894	,028	1,839	1,949
2	1,544	,028	1,488	1,599

2. AGGRESS

Misura: MEASURE_1

AGGRESS	Media	Errore std.	Intervallo di confidenza 95%	
			Limite inferiore	Limite superiore
1	1,737	,025	1,687	1,787
2	1,738	,026	1,687	1,790
3	1,740	,027	1,687	1,792
4	1,661	,027	1,607	1,714

Tabella 6. Risultati dell'analisi del trend


Test dei contrasti entro soggetti

Misura: MEASURE_1

Sorgente	AGGRESS	Somma dei quadrati Tipo III	df	Media dei quadrati	F	Sig.	Eta quadrato parziale	Non centralità	Potenza osservata ^a
AGGRESS	Lineare	,537	1	,537	5,108	,025	,024	5,108	,614
	Quadratico	,337	1	,337	4,800	,030	,023	4,800	,587
	Cubico	,066	1	,066	,954	,330	,005	,954	,163
AGGRESS * SEX	Lineare	,035	1	,035	,332	,565	,002	,332	,088
	Quadratico	,246	1	,246	3,508	,062	,017	3,508	,462
	Cubico	,117	1	,117	1,681	,196	,008	1,681	,252
Errore(AGGRESS)	Lineare	21,659	206	,105					
	Quadratico	14,449	206	,070					
	Cubico	14,323	206	,070					

a. Calcolato usando alfa = ,05

Figura 3. Grafico delle medie attraverso i diversi livelli della variabile entro i soggetti



3. L'ANOVA con il linguaggio SINTASSI

Di seguito presentiamo i programmi SINTASSI per le analisi descritte nel testo. Tutti i disegni che prevedono solo fattori tra i soggetti possono essere analizzati sia con la procedura UNIANOVA, sia con la procedura GLM. Di seguito utilizzeremo la procedura UNIANOVA perché è quella utilizzata dal programma quando le analisi vengono effettuate tramite menù.

Illustriamo i comandi specificati nella procedura UNIANOVA riportata di seguito, per effettuare l'analisi della varianza a una via (o disegno unifattoriale) "tra i soggetti".

*** Analisi della varianza: Disegno Unifattoriale (ad una via)
"Between-subjects": modello "base".**

```
UNIANOVA
  risolti BY condizio
  /METHOD = SSTYPE(3)
  /INTERCEPT = INCLUDE
  /PLOT = PROFILE( condizio )
  /EMMEANS = TABLES(condizio)
  /PRINT = HOMOGENEITY DESCRIPTIVE ETASQ OPOWER
  /CRITERIA = ALPHA(.05)
  /DESIGN = condizio .
```

Il comando `UNIANOVA risolti BY condizio` specifica le variabili che vengono sottoposte ad analisi: in particolare a sinistra di "BY" viene indicata la variabile dipendente mentre a destra di "BY" vengono indicate le indipendenti (in questo caso abbiamo una sola indipendente perché il modello è a una via).

Il comando `/METHOD = SSTYPE(3)` serve a specificare il metodo utilizzato per calcolare la devianza residua. Tratteremo questo aspetto in modo più approfondito nel paragrafo 4.1 di questo documento.

Il comando `/INTERCEPT = INCLUDE` serve per includere l'intercetta (o grand mean) nella stima dei parametri del modello lineare. Di fatto si tratta di una opzione che il programma considera di default.

Il comando `/PLOT = PROFILE(condizio)` serve per ottenere nell'output la rappresentazione grafica (*plot*) delle medie dei gruppi rispetto al fattore specificato in parentesi.

Il comando `/EMMEANS = TABLES(condizio)` serve per ottenere nell'output la tabella che contiene le medie marginali stimate dei gruppi rispetto al fattore specificato in parentesi.

Il comando `/PRINT = HOMOGENEITY DESCRIPTIVE ETASQ OPOWER` consente di ottenere nell'output rispettivamente: a) il risultato del test di Levene per verificare l'omogeneità delle varianze dei gruppi (HOMOGENEITY); b) le statistiche descrittive ovvero la media, la

deviazione standard e il numero di casi per la variabile dipendente nei diversi gruppi

(DESCRIPTIVE); c) il coefficiente eta-quadrato per quantificare l'ampiezza dell'effetto (ETASQ);

d) la stima della potenza della verifica (OPOWER);

Il comando `/CRITERIA = ALPHA(.05)` specifica il livello di probabilità (alpha) utilizzato per esaminare la significatività degli effetti.

Il comando `/DESIGN = condizio` specifica gli effetti che verranno esaminati nel disegno fattoriale: di fatto in questo caso abbiamo una sola variabile indipendente, quindi verrà esaminato solo l'effetto principale di tale variabile .

*** Analisi della varianza: Disegno Unifattoriale (ad una via)
"Between-subjects": confronti post-hoc.**

```
UNIANOVA
  risolti BY condizio
  /METHOD = SSTYPE(3)
  /INTERCEPT = INCLUDE
  /POSTHOC = condizio ( TUKEY )
  /PLOT = PROFILE( condizio )
  /EMMEANS = TABLES(condizio)
  /PRINT = HOMOGENEITY
  /CRITERIA = ALPHA(.05)
  /DESIGN = condizio .
```

Rispetto alla sintassi specificata nel modello "base" l'unico comando nuovo è rappresentato da

`/POSTHOC = condizio (TUKEY)` che serve a specificare che le medie relative ai gruppi relativi ai livelli del fattore "condizio" verranno confrontati utilizzando il metodo HSD di Tukey per l'esame dei confronti post-hoc.

*** Analisi della varianza unifattoriale "Between-subjects":
contrasti pianificati con CONTRAST.**

```
UNIANOVA
  risolti BY condizio
  /CONTRAST (condizio)=special (-.5 -.5 1)
  /CONTRAST (condizio)=special (-1 1 0)
/METHOD = SSTYPE(3)
  /INTERCEPT = INCLUDE
  /PRINT = DESCRIPTIVE ETASQ OPOWER
  /CRITERIA = ALPHA(.05)
  /DESIGN = condizio .
```

Rispetto alla sintassi specificata nel modello "base" l'unico comando nuovo è rappresentato da:

```
/CONTRAST (condizio)=special (-.5 -.5 1)
```

```
/CONTRAST (condizio)=special (-1 1 0)
```

Si tratta di due comandi che utilizzati in combinazione definiscono un set di confronti pianificati ortogonali tramite i quali è possibile confrontare: a) le medie dei primi due gruppi accorpati rispetto alla media del terzo gruppo (con il primo "/CONTRAST"); b) la media del primo gruppo rispetto alla media del secondo gruppo (con il secondo "/CONTRAST").

Per effettuare lo stesso set di confronti è possibile utilizzare una sintassi alternativa tramite il comando "/matrix", come specificato qui di seguito.

*** Analisi della varianza unifattoriale "Between-subjects":
contrasti pianificati con lmatrix.**

```
UNIANOVA
  risolti BY condizio
  /lmatrix condizio -.5 -.5 1
  /lmatrix condizio -1 1 0
/METHOD = SSTYPE(3)
  /INTERCEPT = INCLUDE
  /PRINT = DESCRIPTIVE ETASQ OPOWER
  /CRITERIA = ALPHA(.05)
  /DESIGN = condizio .
```

Illustriamo ora i comandi specificati nella procedura UNIANOVA riportata di seguito, per effettuare l'analisi della varianza fattoriale "tra i soggetti".

```
* Analisi della varianza fattoriale "Between-subjects":  
modello "base".
```

```
UNIANOVA  risposte  BY  rinforzo  compito  
/METHOD = SSTYPE(3)  
/INTERCEPT = INCLUDE  
/PRINT = DESCRIPTIVE ETASQ OPOWER  
/PLOT = PROFILE( rinforzo compito rinforzo*compito )  
  /EMMEANS = TABLES(rinforzo)  
  /EMMEANS = TABLES(compito)  
  /EMMEANS = TABLES ( rinforzo*compito )  
/CRITERIA = ALPHA(.05)  
/DESIGN = rinforzo compito rinforzo*compito .
```

Il comando UNIANOVA risposte BY rinforzo compito specifica anche in questo caso le variabili che vengono sottoposte ad analisi. Mentre la dipendente è a sinistra di "BY" ed è "risposte", ci sono due differenti variabili indipendenti che vengono specificate a destra di "BY", "rinforzo" e "compito".

Il comando /PLOT = PROFILE(rinforzo compito rinforzo*compito) serve per ottenere nell'output la rappresentazione grafica (*plot*): a) delle medie marginali dei gruppi definiti dai livelli dei due fattori "rinforzo" e "compito"; b) delle celle che vengono a essere definite dall'incrocio dei livelli dei due fattori ("rinforzo*compito").

I comandi /EMMEANS = TABLES(rinforzo) /EMMEANS = TABLES(compito) /EMMEANS = TABLES (rinforzo*compito) servono per ottenere nell'output le tabelle che contengono rispettivamente: a) le medie marginali stimate dei gruppi rispetto al fattore rinforzo; b) le medie marginali stimate dei gruppi rispetto al fattore compito; c) le medie stimate delle celle definite dall'incrocio dei livelli dei due fattori rinforzo e compito.

Il comando `/DESIGN = rinforzo compito rinforzo*compito .` specifica gli effetti che verranno esaminati nel disegno fattoriale, in questo caso sono gli effetti principali dei due fattori e l'interazione tra i 2 fattori.

Gli altri comandi sono identici a quelli considerati nel disegno unifattoriale.

*** Analisi della varianza fattoriale "Between-subjects":
Effetti semplici.**

```
UNIANOVA  risposte  BY  rinforzo  compito
/METHOD = SSTYPE(3)
/INTERCEPT = INCLUDE
/EMMEANS = TABLES(rinforzo*compito)
COMPARE (COMPITO) ADJ(SIDAK)
/PRINT = DESCRIPTIVE ETASQ OPOWER
/CRITERIA = ALPHA(.05)
/DESIGN = rinforzo compito rinforzo*compito .
```

Rispetto alla sintassi specificata nel modello "base" il comando `/EMMEANS = TABLES(rinforzo*compito) COMPARE (COMPITO) ADJ(SIDAK)` consente di effettuare l'analisi degli effetti semplici i cui risultati sono stati discussi nel paragrafo 1 di questo documento (si veda anche pag. 143 del testo). In particolare specificando `COMPARE (COMPITO)` si definisce il fattore all'interno dei livelli del quali verranno effettuati i test univariati che sono propri dell'analisi degli effetti semplici. Invece specificando `ADJ(SIDAK)` viene operata la correzione di Sidak del livello di alfa per limitare l'inflazione della probabilità di commettere errore del primo tipo.

I disegni che prevedono almeno un fattore “entro i soggetti” non possono essere analizzati utilizzando la procedura UNIANOVA ma tramite la procedura GLM. Ne illustriamo di seguito le caratteristiche fondamentali.

*** Analisi della varianza a una via “Entro i soggetti”:
modello base.**

```
GLM
  non_ins neutrale insalien
  /WSFACTOR = stimolo 3
  /METHOD = SSTYPE(3)
  /PLOT = PROFILE( stimolo )
  /EMMEANS = TABLES(stimolo)
  /PRINT = DESCRIPTIVE ETASQ OPOWER
  /CRITERIA = ALPHA(.05)
  /WSDESIGN = stimolo .
```

Il comando `GLM non_ins neutrale insalien` specifica le variabili che vengono sottoposte ad analisi. Diversamente dalla procedura UNIANOVA in GLM può essere specificata più di una variabile dipendente. Nel nostro caso abbiamo specificato tre dipendenti che di fatto rappresentano ciascuna un differente livello dell’unica variabile entro i soggetti che abbiamo, e che verrà specificata nel comando successivo. Poiché è assente il comando “BY”, si tratta di un disegno che non prevede variabili tra i soggetti. Come vedremo nel caso dei disegni misti, le variabili che seguono “BYE” che sono alla sua destra sono indipendenti.

Il comando `/WSFACTOR = stimolo 3` specifica quale è la variabile indipendente “entro i soggetti” e quanti livelli ha. Nel nostro caso si tratta della variabile che viene chiamata “stimolo” e che ha 3 livelli: questi livelli sono rappresentati dai punteggi nelle tre variabili dipendenti “non_ins” “neutrale” “insalien” specificate nel comando GLM. Ovviamente il numero di livelli della variabile indipendente specificato in “WSFACTOR” deve essere uguale al numero di variabili dipendenti specificate in “GLM”.

Il comando `/WSDESIGN = stimolo .` specifica gli effetti che verranno esaminati nel disegno fattoriale: di fatto in questo caso abbiamo una sola variabile indipendente, quindi verrà

esaminato solo l'effetto principale di tale variabile .

Gli altri comandi sono identici a quelli considerati nei disegni tra i soggetti.

*** Analisi della varianza a una via "Entro i soggetti":
Confronti post hoc.**

```
GLM
  non_ins neutrale insalien
  /WSFACTOR = stimolo 3
  /METHOD = SSTYPE(3)
  /PLOT = PROFILE( stimolo )
  /EMMEANS = TABLES(stimolo) COMPARE ADJ(SIDAK)
  /PRINT = DESCRIPTIVE ETASQ OPOWER
  /CRITERIA = ALPHA(.05)
  /WSDESIGN = stimolo .
```

Rispetto alla sintassi specificata nel modello "base" l'unico comando nuovo è rappresentato da
/EMMEANS = TABLES(stimolo) COMPARE ADJ(SIDAK) . Per le variabili "entro i
soggetti" non è possibile utilizzare il comando "/POSTHOC" per specificare appunto i confronti
post-hoc. Al posto di questo comando è possibile utilizzare il comando "/EMMEANS" specificando
COMPARE per confrontare le medie e ADJ(SIDAK) per correggere il livello di significatività.

*** Analisi della varianza a una via "Entro i soggetti":
Confronti pianificati.**

```
GLM
  non_ins neutrale insalien
  /WSFACTOR = stimolo 3 special(1 1 1 .5 -.5 0 -.5 -.5 1)
  /METHOD = SSTYPE(3)
  /PLOT = PROFILE( stimolo )
  /EMMEANS = TABLES(stimolo)
  /PRINT = DESCRIPTIVE ETASQ OPOWER
  /CRITERIA = ALPHA(.05)
  /WSDESIGN = stimolo .
```

Per specificare i confronti pianificati in GLM è possibile utilizzare l'opzione "SPECIAL" nel
comando che definisce il fattore entro i soggetti. Rispetto all'analogia opzione di "UNIANOVA" è

necessario specificare un numero di insiemi di coefficienti pari al numero di livelli della variabile entro i soggetti, nel nostro caso quindi 3 insiemi di coefficienti: a) 1 1 1: sono coefficienti che di fatto non vengono utilizzati per nessun confronto ma che servono a specificare in modo corretto l'ordine della matrice che contiene i coefficienti dei contrasti; b) .5 -.5 0: sono i coefficienti che definiscono il confronto tra la media del primo gruppo e quella del secondo; b) -.5 -.5 1: sono i coefficienti che definiscono il confronto tra le medie del primo e del secondo gruppo aggregate, e quella del terzo gruppo.

```

* Analisi della varianza fattoriale "Entro i soggetti".

GLM
  imm_in imm_nin lib_in lib_nin
  /WSFACTOR = ricordo 2 stimolo 2
  /METHOD = SSTYPE(3)
  /PLOT = PROFILE( ricordo stimolo ricordo*stimolo )
  /CRITERIA = ALPHA(.05)
  /WSDESIGN = ricordo stimolo ricordo*stimolo .

```

Il comando `GLM imm_in imm_nin lib_in lib_nin` specifica le variabili che vengono sottoposte ad analisi. In questo disegno abbiamo inserito quattro variabili dipendenti ciascuna delle quali rappresenta una particolare combinazioni dei due livelli dei due fattori “entro i soggetti” che vengono specificati nel comando “/WSFACTOR”. Come per la precedente analisi, poiché è assente il comando “BY”, si tratta di un disegno che non prevede variabili tra i soggetti.

Il comando `/WSFACTOR = ricordo 2 stimolo 2` specifica quali sono le variabili indipendenti “entro i soggetti” e quanti livelli hanno. E' bene prestare particolare attenzione a questi comandi, poiché SPSS assume che le variabili dipendenti specificate nel comando “GLM” siano originate dalle seguenti combinazioni dei livelli dei due fattori “entro”:

- a) prima variabile: combinazione del livello 1 del fattore_1 con il livello 1 del fattore_2;
- b) seconda variabile: combinazione del livello 1 del fattore_1 con il livello 2 del fattore_2;

c) terza variabile: combinazione del livello 2 del fattore_1 con il livello 1 del fattore_2;

d) quarta variabile: combinazione del livello 2 del fattore_1 con il livello 2 del fattore_2.

Se ci sono più fattori “entro” il numero di variabili dipendenti specificate nel comando GLM deve essere uguale al prodotto dei livelli dei fattori specificati in “/WSFACTOR”.

Il comando `/WSDSIGN = ricordo stimolo ricordo*stimolo .` specifica gli effetti che verranno esaminati nel disegno fattoriale, in questo caso sono gli effetti principali dei due fattori e l’interazione tra i 2 fattori.

Gli altri comandi sono identici a quelli considerati in precedenza.

*** Analisi della varianza fattoriale: modello fattoriale “misto”.**

```
GLM
  non_ins insalien BY priming
  /WSFACTOR = stimolo 2
  /METHOD = SSTYPE(3)
  /CRITERIA = ALPHA(.05)
  /WSDSIGN = stimolo
  /DESIGN = priming .
```

Nel comando `GLM non_ins insalien BY priming` vengono specificate due variabili dipendenti e una variabile indipendente “tra i soggetti” (è quella che segue “BY”). Le due variabili dipendenti rappresentano i livelli del fattore “entro i soggetti” che viene specificato tramite il comando “/WSFACTOR”. Si tratta dunque di un disegno che incrocia un fattore “tra i soggetti” con un fattore “entro i soggetti” e che per questo viene definito “misto”.

Il comando `/WSDSIGN = stimolo.` specifica gli effetti che verranno esaminati nel disegno fattoriale relativamente ai fattori “entro i soggetti”: anche se viene specificato soltanto il fattore “stimolo” nel disegno viene considerata anche l’interazione con il fattore “priming”.

Il comando `/DESIGN = priming.` specifica gli effetti che verranno esaminati nel disegno fattoriale relativamente ai fattori “tra i soggetti”: in questo caso si tratta solo dell’effetto principale di “priming”.

Gli altri comandi sono identici a quelli considerati in precedenza.

*** Analisi della covarianza fattoriale: verifica dell’assunzione di indipendenza tra covariata e variabili indipendenti.**

```
UNIANOVA
  inte BY marcpast WITH cpa
  /METHOD = SSTYPE(3)
  /INTERCEPT = INCLUDE
  /CRITERIA = ALPHA(.05)
  /DESIGN = marcpast cpa cpa*marcpast .
```

Il comando `UNIANOVA inte BY marcpast WITH cpa` specifica le variabili che vengono sottoposte ad analisi: l’unica novità rispetto a quanto visto nel caso dei disegni “tra i soggetti” è rappresentato dall’opzione “WITH”, alla destra della quale si collocano tutte le variabili che vengono considerate come covariate.

Il comando `/DESIGN = marcpast cpa cpa*marcpast` specifica che oltre all’effetto principale del fattore “marcpast” e oltre l’effetto della covariata “cpa” viene esaminata anche l’interazione “cpa*marcpast”: un’assunzione dell’ANCOVA è infatti che questa interazione non sia significativa.

*** Analisi della covarianza fattoriale: modello base**

```
UNIANOVA
  inte BY marcpast WITH cpa
  /METHOD = SSTYPE(3)
  /INTERCEPT = INCLUDE
  /CRITERIA = ALPHA(.05)
  /DESIGN = cpa marcpast .
```

Di fatto, l'unica differenza tra questo modello e il precedente è l'assenza dell'interazione tra fattore "tra i soggetti" e covariata nel comando "/DESIGN".

4. Opzioni aggiuntive in GLM

4.1. Modello

Nella finestra di dialogo "Modello" sono presenti diverse opzioni, come si può osservare dalla figura 4. Con *Specifica Modello* si può scegliere tra un modello *fattoriale completo* (opzione di default) che contiene tutti gli effetti principali dei fattori, tutti gli effetti principali delle covariate e tutte le interazioni tra i fattori, ma non contiene le interazioni tra fattori e covariate o tra covariate. L'opzione *Personalizzato* consente di specificare un modello che contiene soltanto gli effetti principali e le interazioni desiderate, come ad esempio nel caso dell'analisi della covarianza dove questa opzione viene scelta per poter esaminare la significatività tra covariate e fattori. In questo pannello il simbolo (F) viene utilizzato per segnalare i fattori fissi, il simbolo (C) viene utilizzato per segnalare le covariate, e il simbolo (R) viene utilizzato per segnalare un fattore casuale.

La sezione *Somma dei Quadrati* serve per specificare il metodo per il calcolo della devianza (*Sum of Squares*) che verrà utilizzato. Quando i modelli (bilanciati o non bilanciati) non hanno celle vuote, il metodo *Tipo III* (che rappresenta l'opzione di default) è il più utilizzato.

Ci sono diversi metodi disponibili per il calcolo delle devianze (vedi Stevens, 1990):

- *Tipo I*. Questo metodo è conosciuto anche come metodo della *scomposizione gerarchica* della devianza. Ogni termine è corretto solamente per il termine che lo precede nel modello. Il metodo Tipo I è comunemente usato nei modelli ANOVA bilanciati in cui tutti gli effetti principali sono specificati prima di ogni effetto di interazione di primo ordine, tutti gli effetti di interazioni di primo ordine sono specificati prima di ogni effetto di interazione di second'ordine e così via.

- *Tipo II*. Questo metodo calcola le devianze di un effetto nel modello, correggendole per tutti gli altri effetti diversi dall'effetto stesso. Il metodo Tipo II è comunemente usato nei modelli ANOVA

bilanciati, nei modelli che prevedono solo effetti principali, nei disegni puramente concatenati (“*nested*”).

- *Tipo III*. E' quello più comunemente usato ed è anche quello che il programma utilizza di default, se non si specifica diversamente. Questo metodo calcola le devianze di un effetto nel disegno, correggendole rispetto ad ogni altro effetto che non lo contiene. Il principale vantaggio del metodo Tipo III risiede nel fatto che le devianze sono invarianti rispetto alle frequenze delle celle. Quindi, questo tipo di devianza è spesso considerato utile per modelli non bilanciati in cui non ci sono dati mancanti nelle celle.

- *Tipo IV*. Questo metodo è indicato per situazioni in cui vi sono celle vuote e viene comunemente utilizzato nei modelli in cui si utilizzano i tipi I e II e in ogni modello bilanciato o modello non bilanciato *con celle vuote*.

Includi l'intercetta nel modello. E' un'opzione che consente di deselezionare l'inclusione dell'intercetta nel modello. Questo è plausibile quando i dati hanno una media uguale a 0 (si dice allora che *passano per l'origine*).

L'opzione *Costruisci Termini* (figura 5) consente di specificare un modello personalizzato, creando le diverse componenti che vengono analizzate. Per i fattori e le covariate selezionate infatti è possibile scegliere tra le seguenti opzioni: *Interazione* (opzione di default), che crea il livello più elevato di interazione di tutte le variabili selezionate; *Effetti Principali*, che crea un effetto principale per ogni variabile selezionata; *Tutti 2-vie*, che crea tutte le possibili interazioni a due vie delle variabili selezionate; *Tutti 3-vie* che crea tutte le possibili interazioni a tre vie delle variabili selezionate; *Tutti 4-vie*, che crea tutte le possibili interazioni a quattro vie delle variabili selezionate; *Tutti 5 vie* che crea tutte le possibili interazioni a cinque vie delle variabili selezionate.

Figura 4. Finestra di dialogo *Modello*





Figura 5. . Finestra di dialogo *Modello*: opzione *Costruisci termini*



4.2. Salva

Tramite la finestra di dialogo Salva (figura 6) è possibile salvare alcune variabili, come ad esempio i residui e gli indici di distanza, che eventualmente possono essere utilizzate per esaminare le assunzioni sui dati nei modi illustrati in questo e in altri capitoli. Le opzioni di questa finestra di dialogo sono le seguenti:

Valori Previsti. Salva i valori previsti dal modello lineare per ogni caso. Si tratta dei punteggi nella variabile dipendenti stimabili tramite l'equazione di base del modello lineare introdotto in questo capitolo. Il programma può salvare anche i i valori previsti non standardizzati e gli errori standard dei valori previsti.

Diagnostici. Si tratta di quelle misure per identificare i casi con combinazioni inusuali di valori per le variabili indipendenti e i casi che possono avere un forte impatto sul modello già introdotte nel capitolo sulla regressione lineare. Sono disponibili la distanza di Cook e i valori di Influenza (Leverage). Per una descrizione di questi indici si veda quanto detto a proposito della procedura "Regressione lineare" (vedi il testo alle pagine 29-38).

Residui. Il residuo non standardizzato è uguale alla differenza tra il valore osservato della variabile dipendente e il valore previsto dal modello lineare. Sono disponibili anche i residui Standardizzati, Studentizzati e Con rimozione. Per una descrizione dei tipi di residui si veda quanto detto a proposito della procedura "Regressione lineare".

Salva in un nuovo file. Crea un file SPSS contenente la matrice di varianze-covarianze dei parametri stimati nel modello.

Figura 6. Finestra di dialogo Salva



Riferimenti bibliografici

Ercolani, A.P., Areni, A., e Mannetti, L. (1990). *La ricerca in psicologia*. Roma: Carocci.